
Chapter 15 Multiple Integration

Useful Tip: If you are reading the electronic version of this publication formatted as a *Mathematica* Notebook, then it is possible to view 3-D plots generated by *Mathematica* from different perspectives. First, place your screen cursor over the plot. Then drag the mouse while pressing down on the left mouse button to rotate the plot.

■ 15.1 Double Integral over a Rectangle

Students should read Section 15.1 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.

Integration can be generalized to functions of two or more variables. As the integral of a single-variable function defines area of a plane region under the curve, it is natural to consider a double integral of a two-variable function that defines volume of a solid under a surface. This definition can be made precise in terms of double Riemann sums where rectangular columns (as opposed to rectangles) are used as building blocks to approximate volume (as opposed to area). The exact volume is then obtained as a limit where the number of columns increases without bound.

■ 15.1.1 Double Integrals and Riemann Sums

Let $f(x, y)$ be a function of two variables defined on a rectangular domain $R = [a, b] \times [c, d]$ in \mathbb{R}^2 . Let $P = \{a = x_0 < x_1 < \dots < x_m = b, c = y_0 < y_1 < \dots < y_n = d\}$ be an arbitrary partition of R into a grid of $m \cdot n$ rectangles, where m and n are integers. For each sub-rectangle $R_{ij} = [x_{i-1}, x_i] \times [y_{j-1}, y_j]$ denote by ΔA_{ij} its area and choose an arbitrary base point $(x_{ij}, y_{ij}) \in R_{ij}$, where $x_{ij} \in [x_{i-1}, x_i]$ and $y_{ij} \in [y_{j-1}, y_j]$. The product $f(x_{ij}, y_{ij}) \Delta A_{ij}$ represents the volume of the ij -rectangular column situated between the surface and the xy -plane. We then define the *double Riemann sum* S_P of $f(x, y)$ on R with respect to P to be the total volume of all these columns:

$$S_P = \sum_{i=1}^m \sum_{j=1}^n f(x_{ij}, y_{ij}) \Delta A_{ij}$$

Define $\|P\|$ to be the maximum dimension of all the sub-rectangles. The *double integral* of $f(x, y)$ on the rectangle R is then defined as the limit of S_P as $\|P\| \rightarrow 0$:

$$\int \int_R f(x, y) dA = \lim_{\|P\| \rightarrow 0} \sum_{i=1}^m \sum_{j=1}^n f(x_{ij}, y_{ij}) \Delta A_{ij}$$

If the limit exists regardless of the choice of partition and base points, then the double integral is said to exist. Otherwise, the double integral does not exist.

MIDPOINT RULE (Uniform Partitions): Let us consider uniform partitions P , where the points $\{x_i\}$ and $\{y_j\}$ are evenly spaced, that is, $x_i = a + i \Delta x$, $y_j = b + j \Delta y$ for $i = 0, 1, \dots, m$ and $j = 0, 1, \dots, n$, and with $\Delta x = (b - a)/m$ and $\Delta y = (d - c)/n$. Then the corresponding double Riemann sum is

$$S_{m,n} = \sum_{i=1}^m \sum_{j=1}^n f(x_{ij}, y_{ij}) \Delta x \Delta y$$

Here is a subroutine called **MDOUBLERSUM** that calculates the double Riemann sum $S_{m,n}$ of $f(x, y)$ over a rectangle R for uniform partitions using the center midpoint of each sub-rectangle as base point, that is, $x_{ij} = (x_{i-1} + x_i)/2 = a + (i - 1/2) \Delta x$ and

$$y_{ij} = (y_{j-1} + y_j) / 2 = c + (j - 1/2) \Delta y.$$

```
In[435]:= Clear[f]
MDOUBLERSUM[a_, b_, c_, d_, m_, n_] :=
Sum[f[a + (i - 1/2) * (b - a) / m, c + (j - 1/2) * (d - c) / n] * (b - a) / m * (d - c) / n,
{i, 1, m}, {j, 1, n}]
```

Example 15.1. Approximate the volume of the solid bounded below the surface $f(x) = x^2 + y^2$ and above the rectangle $R = [-1, 1] \times [-1, 1]$ on the xy -plane using a uniform partition with $m = 10$ and $n = 10$ and center midpoints as base points. Then experiment with larger values of m and n and conjecture an answer for the exact volume.

Solution: We calculate the approximate volume for $m = 10$ and $n = 10$ using the subroutine MDOUBLERSUM:

```
In[437]:= f[x_, y_] = x^2 + y^2;
MDOUBLERSUM[-1, 1, -1, 1, 10, 10]
```

```
Out[438]= 66
25
```

```
In[439]:= N[%]
```

```
Out[439]= 2.64
```

```
In[440]:= Table[MDOUBLERSUM[-1, 1, -1, 1, 10 * k, 10 * k], {k, 1, 10}]
```

```
Out[440]= { 66/25, 133/50, 1798/675, 533/200, 1666/625, 3599/1350, 3266/1225, 2133/800, 16198/6075, 3333/1250 }
```

```
In[441]:= N[%]
```

```
Out[441]= {2.64, 2.66, 2.6637, 2.665, 2.6656, 2.66593, 2.66612, 2.66625, 2.66634, 2.6664}
```

It appears that the exact volume is $8/3$. To prove this, we evaluate the double Riemann sum $S_{m,n}$ in the limit as $m, n \rightarrow \infty$:

```
In[442]:= Clear[s, m, n];
S[m_, n_] = Simplify[MDOUBLERSUM[-1, 1, -1, 1, m, n]]
```

```
Out[443]= 4/3 (2 - 1/m^2 - 1/n^2)
```

```
In[444]:= Limit[Limit[S[m, n], m -> Infinity], n -> Infinity]
```

```
Out[444]= 8
3
```


To see this limiting process visually, evaluate the following subroutine, called **DOUBLEMIDPT**, which plots the surface of the function corresponding to the double integral along with the rectangular columns defined by the double Riemann sum considered in the previous subroutine **MDOUBLERSUM**.

```

In[445]:= Clear[f];
DOUBLEMIDPT[f_, {a_, b_, m_}, {c_, d_, n_}] := Module[
  {dx, dy, i, j, xstar, ystar, mrect, plot},
  dx = N[(b - a) / m];
  xstar = Table[a + i * dx, {i, 0, m}];
  dy = N[(d - c) / n];
  ystar = Table[c + j * dy, {j, 0, n}];
  mcolumn = Table[Cuboid[{xstar[[i]], ystar[[j]], 0},
    {xstar[[i + 1]], ystar[[j + 1]], f[(xstar[[i]] + xstar[[i + 1]]) / 2,
      (ystar[[j]] + ystar[[j + 1]]) / 2]}], {i, 1, m}, {j, 1, n}];
  plot = Plot3D[f[x, y], {x, a, b}, {y, c, d}, Filling -> Bottom];
  Show[plot, Graphics3D[mcolumn], ImageSize -> {300}]

In[447]:= f[x_, y_] := x^2 + y^2;
DOUBLEMIDPT[f, {-1, 1, 10}, {-1, 1, 10}]


```



Here is an animation that demonstrates how the volume of the rectangular columns approach that of the solid in the limit as $m, n \rightarrow \infty$:

Important Note: If you are reading the printed version of this publication, then you will not be able to view any of the animations generated from the **Animate** command in this chapter. If you are reading the electronic version of this publication formatted as a *Mathematica* Notebook, then evaluate each **Animate** command to view the corresponding animation. Just click on the arrow button to start the animation. To control the animation just click at various points on the sliding bar or else manually drag the bar.

```
In[449]:= Animate[DOUBLEMIDPT[f, {-1, 1, a}, {-1, 1, a}], {a, 5, 50, 5}]
```



■ 15.1.2 Double Integrals and Iterated Integrals in *Mathematica*

The *Mathematica* command for evaluating double integrals is the same as that for evaluating integrals of a single-variable function, except that two limits of integration must be specified, one for each independent variable. Thus:

Integrate[f[x,y],{x,a,c},{y,c,d}] analytically evaluates the double integral $\iint_R f(x, y) dA$ over the rectangle $R = [a, b] \times [c, d]$.

NIntegrate[f[x,y],{x,a,c},{y,c,d}] numerically evaluates the double integral $\iint_R f(x, y) dA$ over the rectangle $R = [a, b] \times [c, d]$.

Iterated Integrals: In practice, one does not actually use the limit definition in terms of Riemann sums to evaluate double integrals, but instead apply **Fubini's Theorem** to easily compute them in terms of iterated integrals:

Fubini's Theorem: (Rectangular Domains) If $R = \{(x, y) : a \leq x \leq b, c \leq y \leq d\}$, then

$$\iint_R f(x, y) dA = \int_a^b \int_c^d f(x, y) dy dx = \int_c^d \int_a^b f(x, y) dx dy$$

Thus, *Mathematica* will naturally apply Fubini's Theorem whenever possible to analytically determine the answer. Depending on the form of the double integral, *Mathematica* may resort to more sophisticated integration techniques, such as contour integration, which are beyond the scope of this text.

Example 15.2. Calculate the volume of the solid bounded below by the surface $f(x) = x^2 + y^2$ and above the rectangle $R = [-1, 1] \times [-1, 1]$.

Solution: The volume of the solid is given by the double integral $\iint_R f(x, y) dA$. To evaluate it, we use the **Integrate** command:

```
In[450]:= f[x_, y_] := x^2 + y^2;
Integrate[f[x, y], {x, -1, 1}, {y, -1, 1}]
```

```
Out[451]= 8/3
```

This confirms the conjecture that we made in the previous example for the exact volume.

NOTE: Observe that we obtain the same answer by explicitly computing this double integral as an integrated integral as follows. Moreover, for rectangular domains, the order of integration does not matter.

```
In[452]:= Integrate[Integrate[f[x, y], {x, -1, 1}], {y, -1, 1}]
Integrate[Integrate[f[x, y], {y, -1, 1}], {x, -1, 1}]
```

```
Out[452]= 8/3
```

```
Out[453]= 8/3
```

Example 15.3. Compute the double integral $\iint_R x e^{-y^2} dA$ on the rectangle $R = [0, 1] \times [0, 1]$.

Solution: Observe that the **Integrate** command here gives us an answer in terms of the non-elementary error function **Erf**:

```
In[454]:= Integrate[x * E^(-y^2), {x, 0, 1}, {y, 0, 1}]
```

```
Out[454]= 1/4 * Sqrt[Pi] * Erf[1]
```

This is because the function $f(x, y) = x e^{-y^2}$ has no elementary anti-derivative with respect to y due to the Gaussian factor e^{-y^2} (bell curve). Thus, we instead use the **NIntegrate** Command to numerically approximate the double integral:

```
In[455]:= NIntegrate[x * E^(-y^2), {x, 0, 1}, {y, 0, 1}]
```

```
Out[455]= 0.373412
```

■ Exercises

- Consider the function $f(x, y) = 16 - x^2 - y^2$ defined over the rectangle $R = [0, 2] \times [-1, 3]$.
 - Use the subroutine **MDOUBLERSUM** to compute the double Riemann sum $S_{m,n}$ of $f(x, y)$ over R for $m = 2$ and $n = 2$.
 - Repeat part a) by generating a table of double Riemann sums for $m = 10k$ and $n = 10k$ where $k = 1, 2, \dots, 10$. Make a conjecture for the exact value of $\iint_R f(x, y) dA$.
 - Find a formula for $S_{m,n}$ in terms of m and n . Verify your conjecture in part b) by evaluating $\lim_{m,n \rightarrow \infty} S_{m,n}$.

- d. Directly compute $\iint_R f(x, y) dA$ using the **Integrate** command.
2. Repeat Exercise 1 but with $f(x, y) = (1 + x)(1 + y)(1 + xy)$ defined over the rectangle $R = [0, 1] \times [0, 1]$.
3. Evaluate the double integral $\iint_R \sqrt{x^4 + y^4} dA$ over the rectangle $R = [-2, 1] \times [-1, 2]$ using both the **Integrate** and **NIntegrate** commands. How do the two answers compare?
4. Calculate the volume of the solid lying under the surface $z = e^{-y}(x + y^2)$ and over the rectangle $R = [0, 2] \times [0, 3]$. Then make a plot of this solid.
5. Repeat Exercise 4 but with $z = \sin(x^2 + y^2)$ and rectangle $R = [-\sqrt{\pi}, \sqrt{\pi}] \times [-\sqrt{\pi}, \sqrt{\pi}]$.
6. Evaluate the double integral $\iint_R f(x, y) dA$ where $f(x, y) = xy \cos(x^2 + y^2)$ and $R = [-\pi, \pi] \times [-\pi, \pi]$. Does your answer make sense? Make a plot of the solid corresponding to this double integral to intuitively explain your answer. HINT: Consider symmetry.
7. Find the volume of solid bounded between the two hyperbolic paraboloids (saddles) $z = 1 + x^2 - y^2$ and $z = 3 - x^2 + y^2$ over the rectangle $R = [-1, 1] \times [-1, 1]$.
8. Find the volume of the solid bounded by the planes $z = 2x$, $z = -3x + 2$, $y = 0$, $y = 1$, and $z = 0$.

■ 15.2 Double Integral over More General Regions

Students should read Section 15.2 of Rogawski's Calculus [1] for a detailed discussion of the material presented in this section.

For domains of integration that are non-rectangular but still *simple*, that is, bounded between two curves, Fubini's Theorem continues to hold. There are two types to consider:

Fubini's Theorem: (Simple Domains)

Type I (Vertically Simple): If $D = \{(x, y) : a \leq x \leq b, \alpha(x) \leq y \leq \beta(x)\}$, then

$$\iint_D f(x, y) dA = \int_a^b \int_{\alpha(x)}^{\beta(x)} f(x, y) dy dx$$

The corresponding *Mathematica* command is **Integrate[f[x,y],{x,a,b},{y,α[x],β[x]}]**.

Type II (Horizontally Simple): If $D = \{(x, y) : c \leq y \leq d, \alpha(y) \leq x \leq \beta(y)\}$, then

$$\iint_D f(x, y) dA = \int_c^d \int_{\alpha(y)}^{\beta(y)} f(x, y) dx dy$$


The corresponding *Mathematica* command is **Integrate[f[x,y],{y,c,d},{x,α[y],β[y]}]**.

Warning: Be careful not to reverse the order of integration prescribed for either type. For example, evaluating the command **Integrate[f[x,y],{y,α[x],β[x]},{x,a,b}]** for Type I (x and y are reversed) will lead to incorrect results.

Example 15.4. Calculate the volume of the solid bounded below by the surface $f(x, y) = 1 - x^2 + y^2$ and above the domain D bounded by $x = 0$, $x = 1$, $y = x$, and $y = 1 + x^2$.


Solution: We observe that $x = 0$ and $x = 1$ represent the left and right boundaries, respectively, of D . Therefore, we plot the graphs of the other two equations along the x -interval $[0, 2]$ to visualize D (shaded in the following plot):

```
In[456]:= Clear[x, y]
plot1 = Plot[{x, 1 + x^2}, {x, 0, 1}, Filling -> {1 -> {2}}, ImageSize -> 250]
```



Here is a plot of the corresponding solid situated over D :

```
In[458]:= f[x_, y_] = 1 - x^2 + y^2;
plot3 = Plot3D[f[x, y], {x, 0, 1}, {y, x, 1 + x^2}, Filling -> Bottom,
ViewPoint -> {1, 1, 1}, PlotRange -> {0, 4}, ImageSize -> {250}]
```



To compute the volume of this solid given by $\iint_D f(x, y) dA$, we describe D as a vertically simple domain where $0 \leq x \leq 1$ and $x \leq y \leq 1 + x^2$ and apply Fubini's Theorem to evaluate the corresponding iterated integral $\int_0^1 \int_x^{1+x^2} f(x, y) dy dx$ (remember to use the correct order of integration):


```
In[460]:= Integrate[f[x, y], {x, 0, 1}, {y, x, 1 + x^2}]
```

Out[460]= $\frac{29}{21}$

Example 15.5. Evaluate the double integral $\iint_D \sin(y^2) dA$, where D is the domain bounded by $x = 0$, $y = 2$, and $y = x$.


Solution: We first plot the graphs of $x = 0$, $y = 2$, and $y = x$ to visualize the domain D :

```
In[461]:= plot1 = ContourPlot[{x == 0, y == 2, y == x},
  {x, -0.5, 2.5}, {y, -0.5, 2.5}, ImageSize -> {250}]
```



It follows that D is the triangular region bounded by these graphs, which we shade in the following plot to make clear:

```
In[462]:= plot2 = Plot[x, {x, 0, 2}, Filling -> 2];
Show[plot1, plot2, ImageSize -> {250}]
```



To compute the given double integral, we describe D as a horizontally simple domain, where $0 \leq y \leq 2$ and $0 \leq x \leq y$ and apply Fubini's Theorem to evaluate the corresponding iterated integral $\int_0^2 \int_0^y \sin(y^2) dx dy$ (again, remember to use the correct order of integration):

```
In[464]:= Integrate[Sin[y^2], {y, 0, 2}, {x, 0, y}]
```

Out[464]= $\sin[2]^2$

```
In[465]:= N[%]
```

Out[465]= 0.826822

NOTE: It is also possible to view D as a vertically simple domain, where $0 \leq x \leq 2$ and $x \leq y \leq 2$. The corresponding iterated

integral $\int_0^2 \int_x^2 \sin(y^2) dy dx$ gives the same answer, as it should by Fubini's Theorem:

```
In[466]:= Integrate[Sin[y^2], {x, 0, 2}, {y, x, 2}]
```

```
Out[466]:= Sin[2]^2
```

Observe that it is actually impossible to evaluate this iterated integral by hand since there is no elementary formula for the anti-derivative of $\sin(y^2)$ with respect to y . Thus, if necessary, *Mathematica* automatically switches the order of integration by converting from one type to the other.


■ Exercises

In Exercises 1 through 4, evaluate the given iterated integrals and plot the solid corresponding to each one.

1. $\int_0^1 \int_0^{x^2} (4 - x^2 + y^2) dy dx$
2. $\int_0^4 \int_0^{\sqrt{2-y^2}} x^2 y dx dy$
3. $\int_0^\pi \int_0^{\sin \theta} r^2 \cos \theta dr d\theta$
4. $\int_0^1 \int_0^{\frac{x}{1+xy}} dy dx$

In Exercises 5 through 8, evaluate the given double integrals and plot the solid corresponding to each one.

5. $\iint_D (x + y) dA$, $D = \{(x, y) : 0 \leq x \leq 3, 0 \leq y \leq \sqrt{x}\}$
6. $\iint_D \sqrt{x + y} dA$, $D = \{(x, y) : 0 \leq x \leq 1 - y^2, 0 \leq y \leq 1\}$
7. $\iint_D e^{x+y} dA$, where $D = \{(x, y) : x^2 + y^2 \leq 4\}$
8. $\iint_D \frac{y}{x+1} dA$, where D is the following shaded diamond region:



In Exercises 9 through 12, calculate the volume of the given solid S :

9. S is bounded under the paraboloid $z = 16 - x^2 - y^2$ and above the region bounded between the line $y = x$ and the parabola $y = 6 - x^2$.
10. S is bounded under the right circular cone $z = \sqrt{x^2 + y^2}$ and above the disk $x^2 + y^2 \leq 1$.
11. S is bounded between the plane $z = 5 + 2x + 2y$ and the paraboloid $z = 12 - x^2 - y^2$. HINT: Equate the two surfaces to obtain the equation of the domain.
12. S is bounded between the cylinders $x^2 + y^2 = 1$ and $y^2 + z^2 = 1$.

■ 15.3 Triple Integrals

Students should read Section 15.3 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.

Once the notion of a double integral is well established, it is straightforward to generalize it to triple (and even higher-order) integrals for functions of three variables defined over a solid region in space. Here is the definition of a triple integral in terms of

triple Riemann sums for a function $f(x, y, z)$ defined on a box region $B = \{(x, y, z) : a \leq x \leq b, c \leq y \leq d, p \leq z \leq q\}$ (refer to your calculus text for details):

$$\iiint_B f(x, y, z) \, dV = \lim_{\|P\| \rightarrow \infty} \sum_{i=1}^m \sum_{j=1}^n \sum_{k=1}^p f(x_{ijk}, y_{ijk}) \Delta V_{ijk}$$

where the notation is analogous to that used for double integrals in Section 15.1 of this text. Of course, Fubini's Theorem also generalizes to triple integrals:

Fubini's Theorem: (Box Domains) If $B = \{(x, y, z) : a \leq x \leq b, c \leq y \leq d, p \leq z \leq q\}$, then

$$\iiint_B f(x, y, z) \, dV = \int_a^b \int_c^d \int_p^q f(x, y, z) \, dz \, dy \, dx$$

The corresponding *Mathematica* commands are:

Integrate[**f**[**x,y,z**],{**x,a,c**},{**y,c,d**},{**z,e,f**}] analytically evaluates the triple integral $\iiint_B f(x, y, z) \, dV$ over the box $B = [a, b] \times [c, d] \times [e, f]$.

NIntegrate[**f**[**x,y,z**],{**x,a,c**},{**y,c,d**},{**z,e,f**}] numerically evaluates the triple integral $\iiint_B f(x, y, z) \, dV$ over the rectangle $B = [a, b] \times [c, d] \times [e, f]$.

NOTE: For box domains, the order of integration does not matter so that it is possible to write five other versions of triple iterated integrals besides the one given in Fubini's Theorem.

Example 15.6. Calculate the triple integral $\iiint_B x y z \, dV$ over the box $B = [0, 1] \times [2, 3] \times [4, 5]$.

Solution: We use the **Integrate** command to calculate the given triple integral.

In[467]:= **Integrate**[**x y z**, {**x**, 0, 1}, {**y**, 2, 3}, {**z**, 4, 5}]

Out[467]= $\frac{45}{8}$


Volume as Triple Integral: Recall that if a solid region W is bounded between two surfaces $\psi(x, y)$ and $\phi(x, y)$, where both are defined on the same domain D with $\psi(x, y) \leq \phi(x, y)$, then its volume V can be expressed by the triple integral

$$V = \iiint_W 1 \, dV = \iint_D \int_{\psi(x,y)}^{\phi(x,y)} 1 \, dz \, dA$$

Example 15.7. Calculate the volume of the solid bounded between the surfaces $z = 4x^2 + 4y^2$ and $z = 16 - 4x^2 - 4y^2$ on the rectangular domain $[-1, 1] \times [-1, 1]$.

Solution: Here is a plot of the solid:

```
In[468]:= Plot3D[{4 x^2 + 4 y^2, 16 - 4 x^2 - 4 y^2}, {x, -1, 1}, {y, -1, 1},
  Filling -> {1 -> 8, 2 -> 8}, ImageSize -> {250}, ImagePadding -> {{15, 15}, {15, 15}}]
```



The volume of the solid is given by the triple iterated integral $\int_{-1}^1 \int_{-1}^1 \int_{4x^2+4y^2}^{16-4x^2-4y^2} 1 \, dz \, dy \, dx$:

```
In[469]:= Integrate[1, {x, 0, 1}, {y, -1, 0}, {z, 4 x^2 + 4 y^2, 16 - 4 x^2 - y^2}]
```

Out[469]= $\frac{35}{3}$

■ Exercises

In Exercises 1 through 4, evaluate the given iterated integrals:

- $\int_0^1 \int_0^x \int_0^{y^2} (x + y + z) \, dz \, dy \, dx$
- $\int_0^3 \int_0^{\sin y} \int_0^{y+z} x y z \, dx \, dz \, dy$
- $\int_0^\pi \int_0^\theta \int_0^{r \cos \theta} r z^2 \, dz \, dr \, d\theta$
- $\int_0^1 \int_{\sqrt{x}}^1 \int_0^{1-y} z \log(1 + x y) \, dz \, dy \, dx$

In Exercises 5 through 8, evaluate the given triple integrals:

- $\iiint_W (x + y z) \, dV$, where $W = \{(x, y, z) : 0 \leq x \leq 1, 0 \leq y \leq \sqrt{x}, 0 \leq z \leq y^2\}$.
- $\iiint_W \sin y \, dV$, where W lies under the plane $z = 1 + x + y$ and above the triangular region bounded by $x = 0$, $x = 2$, and $y = 3x$.
- $\iiint_W z \, dV$, where W is bounded by the paraboloid $z = 4 - x^2 - y^2$ and $z = 0$.
- $\iiint_W f(x, y, z) \, dV$, where $f(x, y, z) = z^2$ and W is bounded between the cone $z = \sqrt{x^2 + y^2}$ and $z = 9$.
- The triple integral $\int_0^1 \int_{x/2}^{1-x/2} \int_0^{2-x-2y} dz \, dy \, dx$ represents the volume of a solid S . Evaluate this integral. Then make a plot of S and describe it.

10. Midpoint Rule for Triple Integrals:

- Develop a subroutine called **MTRIPLERSUM** to compute the triple Riemann sum of the triple integral $\iiint_B f(x, y, z) \, dV$ over the box domain $B = \{(x, y, z) : a \leq x \leq b, c \leq y \leq d, p \leq z \leq q\}$ for uniform partitions and using the center midpoint of each sub-box as base point. HINT: Modify the subroutine **MDOUBLERSUM** in Section 15.1 of this text.
- Use your subroutine **MTRIPLERSUM** in part a) to compute the triple Riemann sum of $\iiint_B (x^2 + y^2 + z^2)^{3/2} \, dV$ over the box $B = \{(x, y, z) : 0 \leq x \leq 1, 0 \leq y \leq 2, 0 \leq z \leq 3\}$ by dividing B into 48 equal sub-boxes, that is, cubes having side length of $1/2$.

- c. Repeat part b) by dividing B into cubes having side length of $1/4$ and more generally into cubes having side length of $1/2^n$ for n sufficiently large in order to obtain an approximation accurate to 2 decimal places.
- d. Verify your answer in part c) using *Mathematica's* **NIntegrate** command.

■ 15.4 Integration in Polar, Cylindrical, and Spherical Coordinates

Students should read Section 15.4 of Rogawski's Calculus [1] for a detailed discussion of the material presented in this section.

■ 15.4.1 Double Integrals in Polar Coordinates

The following Change of Variables Formula converts a double integral in rectangular coordinates to one in polar coordinates:

Change of Variables Formula (Polar Coordinates):

I. Polar Rectangles: If $R = \{(r, \theta) : \theta_1 \leq \theta \leq \theta_2, r_1 \leq r \leq r_2\}$, then

$$\iint_R f(x, y) \, dA = \int_{\theta_1}^{\theta_2} \int_{r_1}^{r_2} f(r \cos \theta, r \sin \theta) r \, dr \, d\theta$$

II. Polar Regions: If $D = \{(r, \theta) : \theta_1 \leq \theta \leq \theta_2, \alpha(\theta) \leq r \leq \beta(\theta)\}$, then


$$\iint_D f(x, y) \, dA = \int_{\theta_1}^{\theta_2} \int_{\alpha(\theta)}^{\beta(\theta)} f(r \cos \theta, r \sin \theta) r \, dr \, d\theta$$

Example 15.8. Calculate the volume of the solid region bounded by the paraboloid $f(x) = 4 - x^2 - y^2$ and the xy -plane using polar coordinates.

Solution: We first plot the paraboloid:

```
In[470]:= f[x_, y_] = 4 - x^2 - y^2
Plot3D[f[x, y], {x, -2, 2}, {y, -2, 2}, PlotRange -> {0, 4}, ImageSize -> {250}]
```

```
Out[470]= 4 - x^2 - y^2
```



The circular domain D can be easily described in polar coordinates by the polar rectangle $R = \{(r, \theta) : 0 \leq r \leq 2, 0 \leq \theta \leq 2\pi\}$. Thus, the volume of the solid is given by the corresponding double integral $\int_0^{2\pi} \int_0^2 f(r \cos \theta, r \sin \theta) r \, dr \, d\theta$ in polar coordinates:

```
In[472]:= Clear[r, θ];
          Integrate[r * f[r * Cos[θ], r * Sin[θ]], {r, 0, 2}, {θ, 0, 2 Pi}]
```

```
Out[473]= 8 π
```

Observe that here $f(x, y)$ simplifies nicely in polar coordinates:

```
In[474]:= f[r * Cos[θ], r * Sin[θ]]
          Simplify[%]
```

```
Out[474]= 4 - r^2 Cos[θ]^2 - r^2 Sin[θ]^2
```

```
Out[475]= 4 - r^2
```

NOTE: Evaluating the same double integral in rectangular coordinates by hand would be quite tedious. This is not a problem with *Mathematica*, however:

```
In[476]:= Integrate[f[x, y], {x, -2, 2}, {y, -Sqrt[4 - x^2], Sqrt[4 - x^2]}]
```

```
Out[476]= 8 π
```

■ 15.4.2 Triple Integrals in Cylindrical Coordinates

The following Change of Variables Formula converts a triple integral in rectangular coordinates to one in cylindrical coordinates:


Change of Variables Formula (Cylindrical Coordinates): If a solid region W is described by $\theta_1 \leq \theta \leq \theta_2$, $\alpha(\theta) \leq r \leq \beta(\theta)$, and $z_1(r, \theta) \leq z \leq z_2(r, \theta)$, then

$$\iiint_W f(x, y, z) \, dV = \int_{\theta_1}^{\theta_2} \int_{\alpha(\theta)}^{\beta(\theta)} \int_{z_1(r, \theta)}^{z_2(r, \theta)} f(r \cos \theta, r \sin \theta, z) r \, dz \, dr \, d\theta$$

Example 15.9. Use cylindrical coordinates to calculate the triple integral $\iiint_W z \, dV$, where W is the solid region bounded above by the plane $z = 8 - x - y$, below by the paraboloid $z = 4 - x^2 - y^2$, and inside the cylinder $x^2 + y^2 = 4$.

Solution: Since W lies inside the cylinder $x^2 + y^2 = 4$, this implies that it has a circular base on the xy -plane given by the same equation, which can be described in polar coordinates by $0 \leq \theta \leq 2\pi$ and $0 \leq r \leq 2$. Here is a plot of all three surfaces (plane, paraboloid, and cylinder):

```
In[477]:= plotplane = Plot3D[8 - x - y, {x, -2, 2}, {y, -2, 2}];
plotparaboloid = Plot3D[4 - x^2 - y^2, {x, -2, 2}, {y, -2, 2}];
plotcylinder = ParametricPlot3D[{2 * Cos[θ], 2 * Sin[θ], z}, {θ, 0, 2 π}, {z, 0, 12}];
Show[plotplane, plotparaboloid, plotcylinder, PlotRange → All, ImageSize → {250}]
```



Since W is bounded in z by $4 - x^2 - y^2 \leq z \leq 8 - x - y$, or in cylindrical coordinates, $4 - r \cos \theta - r \sin \theta \leq z \leq 4 - r^2$, it follows that the given triple integral transforms to

$$\int_0^{2\pi} \int_0^2 \int_{4-r^2}^{4-r \cos \theta - r \sin \theta} z r dz dr d\theta$$

Evaluating this integral in *Mathematica* yields the answer

```
In[481]:= Integrate[z * r, {θ, 0, 2 π}, {r, 0, 2},
{z, 4 - r * Cos[θ] - r * Sin[θ], 8 + r * Cos[θ] + r * Sin[θ]}]
```

Out[481]= 96 π

■ 15.4.3 Triple Integrals in Spherical Coordinates

The following Change of Variables Formula converts a triple integral in rectangular coordinates to one in spherical coordinates:

Change of Variables Formula (Spherical Coordinates): If a solid region W is described by $\theta_1 \leq \theta \leq \theta_2$, $\phi_1 \leq \phi \leq \phi_2$, and $\rho_1(\theta, \phi) \leq \rho \leq \rho_2(\theta, \phi)$, then

$$\iiint_W f(x, y, z) dV = \int_{\theta_1}^{\theta_2} \int_{\phi_1}^{\phi_2} \int_{\rho_1(\theta, \phi)}^{\rho_2(\theta, \phi)} f(\rho \cos \theta \sin \phi, \rho \sin \theta \sin \phi, \rho \cos \phi) \rho^2 \sin \phi d\rho d\phi d\theta$$

Example 15.10. Use spherical coordinates to calculate the volume of the solid W lying inside the sphere $x^2 + y^2 + z^2 = z$ and above the cone $z = \sqrt{x^2 + y^2}$.

Solution: In spherical coordinates, the equation of the sphere is given by


$$\rho^2 = \rho \cos \phi$$

or equivalently, $\rho = \cos \phi$. Similarly, the equation of the cone transforms to

$$\rho \cos \phi = \sqrt{(\rho \cos \theta \sin \phi)^2 + (\rho \sin \theta \sin \phi)^2} = \rho \sin \phi$$

It follows that $\cos \phi = \sin \phi$, or $\phi = \pi/4$. Therefore, the cone makes an angle of 45 degrees with respect to the z -axis, as shown in the following plot along with the top half of the sphere:

```
In[482]:= Clear[ρ]
plotcone = ParametricPlot3D[{ρ Cos[θ] Sin[Pi / 4], ρ Sin[θ] Sin[Pi / 4], ρ Cos[Pi / 4]},
  {θ, 0, 2 Pi}, {ρ, 0, Sqrt[2] / 2}];
plotsphere = ParametricPlot3D[{Cos[φ] Cos[θ] Sin[φ],
  Cos[φ] Sin[θ] Sin[φ], Cos[φ] Cos[φ]}, {θ, 0, 2 Pi}, {φ, 0, Pi / 4}];
Show[plotcone, plotsphere, PlotRange → All, ViewPoint → {1, 1, 1 / 4},
  ImageSize → {250}]
```



It is now clear that the solid W is described by $0 \leq \theta \leq 2\pi$, $0 \leq \phi \leq \pi/4$, and $0 \leq \rho \leq \cos \phi$. Thus, its volume is given by the triple integral

$$\int_0^{2\pi} \int_0^{\pi/4} \int_0^{\cos \phi} \rho^2 \sin \phi \, d\rho \, d\phi \, d\theta$$

which in *Mathematica* evaluates to

```
In[486]:= Integrate[ρ^2 * Sin[φ], {θ, 0, 2 Pi}, {φ, 0, Pi / 4}, {ρ, 0, Cos[φ]}]
```

Out[486]= $\frac{\pi}{8}$

■ Exercises

In Exercises 1 through 4, evaluate the given double integral by converting to polar coordinates:

1. $\int_{-1}^1 \int_{-\sqrt{1-x^2}}^{\sqrt{1-x^2}} (1-x^2-y^2) \, dy \, dx$

2. $\int_0^2 \int_0^{\sqrt{4-x^2}} e^{-(x^2+y^2)} \, dy \, dx$

3. $\iint_D x \log y \, dA$, where D is the annulus (donut-shaped region) with inner radius 1 and outer radius 3.

4. $\iint_D \arctan \frac{y}{x} \, dA$, where D is the region inside the cardioid $r = 1 + \cos t$.

5. Use polar coordinates to calculate the volume of the solid that lies below the paraboloid $z = x^2 + y^2$ and inside the cylinder $x^2 + y^2 = 2y$.

6. Evaluate the triple integral $\int_0^2 \int_0^{\sqrt{4-x^2}} \int_0^{\sqrt{4-x^2-y^2}} (x^2 + y^2) dz dy dx$ by converting to cylindrical coordinates.
7. Use cylindrical coordinates to calculate the triple integral $\iiint_W (x^2 + y^2) dV$, where W is the solid bounded between the two paraboloids $z = x^2 + y^2$ and $z = 8 - x^2 - y^2$.
8. Evaluate the triple integral $\int_{-2}^2 \int_{-\sqrt{4-x^2}}^{\sqrt{4-x^2}} \int_{\sqrt{x^2+y^2}}^{\sqrt{4-x^2-y^2}} (x^2 + y^2 + z^2) dz dy dx$ by converting to spherical coordinates.
9. The solid defined by the spherical equation $\rho = \sin \phi$ is called the *torus*.
- Plot the torus.
 - Calculate the volume of the torus.
10. Ice-Cream Cone: A solid W in the shape of an ice-cream cone is bounded below by the cylinder $z = \sqrt{x^2 + y^2}$ and above by the sphere $x^2 + y^2 + z^2 = 8$. Plot W and determine its volume.

■ 15.5 Applications of Multiple Integrals

Students should read Section 15.5 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.

Mass as Double Integral: Consider a lamina (thin plate) D in \mathbb{R}^2 with continuous mass density $\rho(x, y)$. Then the mass of D is given by the double integral


$$M = \iint_D \rho(x, y) dA$$

where the domain of integration is given by the region that describes the lamina D .

Example 15.11. Calculate the mass of the lamina D bounded between the parabola $y = x^2$ and $y = 4$ with density $\rho(x, y) = y$.

Solution: Here is a plot of the lamina D (shaded):

```
In[487]:= Plot[{x^2, 4}, {x, -2, 2}, ImageSize -> {250}, Filling -> {2 -> {1}}]
```




We can view D as a Type I region described by $-2 \leq x \leq 2$ and $x^2 \leq y \leq 4$. Thus, the mass of the lamina is given by the double integral:

```
In[488]:= Integrate[y, {x, -2, 2}, {y, x^2, 4}]
```

Out[488]= $\frac{128}{5}$

NOTE: Mass of a lamina can also be interpreted as the volume of the solid bounded by its density function over D as shown in the following plot:

```
In[489]:= Plot3D[y, {x, -2, 2}, {y, 0, 4}, RegionFunction -> (#1^2 < #2 < 4 &),
      Filling -> Bottom, Mesh -> None, ImageSize -> 250]
```




Example 15.12. Suppose a circular metal plate D , bounded by $x^2 + y^2 = 9$, has electrical charge density $\rho(x, y) = \sqrt{9 - x^2 - y^2}$. Calculate the total charge of the plate.

Solution: Here is a plot of the metal plate D (shaded):

```
In[490]:= Integrate[y, {x, -2, 2}, {y, x^2, 4}]
```

Out[490]= $\frac{128}{5}$

```
In[491]:= Plot[{Sqrt[9 - x^2], -Sqrt[9 - x^2]}, {x, -3, 3},
      ImageSize -> {250}, Filling -> {2 -> {1}}, AspectRatio -> 1]
```



We shall calculate the total charge of the plate using polar coordinates, which will simplify the corresponding double integral.

Since $\rho(r, \theta) = \sqrt{9 - r^2}$ and D is a simple polar region described by $r = 3$, the total charge is

```
In[492]:= Integrate[Sqrt[9 - r^2] * r, {r, 0, 3}, {theta, 0, 2 Pi}]
```

```
Out[492]= 18 π
```

Mass as Triple Integral: We can extend the notion of mass to a solid region W in \mathbb{R}^3 . Suppose W is bounded between two surfaces $z = \psi(x, y)$ and $z = \phi(x, y)$, where both are defined on the same domain D with $\psi(x, y) \leq \phi(x, y)$, and has density $\rho(x, y, z)$. Then the mass of W can be expressed by the triple integral


$$M = \iiint_W \rho(x, y, z) dV = \iint_D \int_{\psi(x,y)}^{\phi(x,y)} \rho(x, y, z) dz dA$$

Example 15.13. Calculate the mass of the solid region W bounded between the planes $z = 1 - x - y$ and $z = 1 + x + y$ and situated over the triangular domain D bounded by $x = 0$, $y = 0$, and $y = 1 - x$. Assume the density of W is given by $\rho(x, y, z) = 1 + x^2 + y^2$.

Solution: Here is a plot of the solid region W :

```
In[493]:= Plot3D[{1 - x - y, 1 + x + y}, {x, 0, 1}, {y, 0, 1 - x}, ViewPoint -> {1, 1, 1},
  Filling -> {1 -> 1, 2 -> 1}, Ticks -> {Automatic, Automatic, {1, 2}},
  ImageSize -> {250}, ImagePadding -> {{15, 15}, {15, 15}}]
```

```
Out[493]=
```



The mass of the solid is given by the triple iterated integral $\int_0^1 \int_0^{1-x} \int_{1-x-y}^{1+x+y} (1 + x^2 + y^2) dz dy dx$:

```
In[494]:= Integrate[1 + x^2 + y^2, {x, 0, 1}, {y, 0, 1 - x}, {z, 1 - x - y, 1 + x + y}]
```

```
Out[494]= 14/15
```

Center of Mass: Given a lamina D in \mathbb{R}^2 , its *center of mass* $(x_{\text{CM}}, y_{\text{CM}})$ (or balance point) is defined as the ratio of its moments (with respect to the coordinate axes) to its mass:

$$x_{\text{CM}} = \frac{M_y}{M}, \quad y_{\text{CM}} = \frac{M_x}{M}$$

where the moments M_y and M_x are defined by

$$M_y = \frac{1}{A} \iint_D x \rho(x, y) dA, \quad M_x = \frac{1}{A} \iint_D y \rho(x, y) dA$$

NOTE: In case the lamina has uniform density, that is, $\rho(x, y) = 1$, then the center of mass is the same as the *centroid* whose coordinates represent averages of the coordinates over the lamina.

Center of mass (and centroid) can be naturally extended to solid objects in \mathbb{R}^3 . Refer to your textbook for further details.

Example 15.14. Calculate the mass of the solid region W bounded between the planes $z = 1 - x - y$ and $z = 1 + x + y$ and situated over the triangular domain D bounded by $x = 0$, $y = 0$, and $y = 1 - x$. Assume the density of W is given by $\rho(x, y, z) = 1 + x^2 + y^2$.

■ Exercises

In Exercises 1 and 2, find the mass of the given lamina D .

1. D is bounded between $y = \sin(\pi x)$ and $y = 0$ along the interval $[0, 1]$ and has density $\rho(x, y) = x(1 - x)$.
2. D is bounded by the lines $y = x + 1$, $y = -2x - 2$, and $x = 1$ and has density $\rho(x, y) = 1 + y^2$.

3. Find the center of mass of the lamina D in Exercises 1 and 2.

4. Find the centroid of the lamina in Exercises 1 and 2. Compare the centroid of each lamina with its center of mass.

In Exercises 5 and 6, find the mass of the given solid object W .

5. W is the interior of the tetrahedron enclosed by the planes $x = 0$, $y = 0$, $z = 0$, and $z = 1 - x - y$ and has density $\rho(x, y, z) = 1 - z$.

6. W is the ice-cream cone bounded below by the cylinder $z = \sqrt{x^2 + y^2}$ and above by the sphere $x^2 + y^2 + z^2 = 8$ and has density $\rho(x, y, z) = z^2$.

7. Find the center of mass of the tetrahedron in Exercises 5 and 6. Refer to your textbook for appropriate formulas.

8. Find the centroid of the tetrahedron in Exercises 5 and 6. Compare this with its center of mass. Refer to your textbook for appropriate formulas.

■ 15.6 Change of Variables

Students should read Section 15.6 of Rogawski's *Calculus* [1] for a detailed discussion of the material presented in this section.

A change of variables is often useful for simplifying integrals of a single variable (commonly referred to as u -substitution):

$$\int_a^b f(x) dx = \int_c^d f(g(u)) g'(u) du$$

where $x = g(u)$, $a = g(c)$, and $b = g(d)$. This substitution formula allows one to transform an integral in the variable x to one in a new variable u . Observe that the interval $[c, d]$ is mapped to interval $[a, b]$ under the function g .

This technique can be extended to double integrals of the form $\iint_D f(x, y) dx dy$, where a change of variables is described by a transformation $G(u, v) = (x, y)$, which maps a region D_0 in the uv -coordinate plane to the region D in the xy -coordinate plane.

The following Change of Variables Formula converts a double integral from the xy -coordinate system to a new coordinate system defined by u and v :

Change of Variables Formula (Coordinate Transformation): If $G(u, v) = (x(u, v), y(u, v))$ is a C^1 -mapping from D_0 to D , then

$$\iint_D f(x, y) dx dy = \iint_{D_0} f(x(u, v), y(u, v)) \left| \frac{\partial(x, y)}{\partial(u, v)} \right| du dv$$

where $\left| \frac{\partial(x,y)}{\partial(u,v)} \right|$, referred to as the *Jacobian* of G and also denoted by $\text{Jac}(G)$, is given by

$$\text{Jac}(G) = \left| \frac{\partial(x,y)}{\partial(u,v)} \right| = \left| \begin{array}{cc} \frac{\partial x}{\partial u} & \frac{\partial x}{\partial v} \\ \frac{\partial y}{\partial u} & \frac{\partial y}{\partial v} \end{array} \right| = \frac{\partial x}{\partial u} \frac{\partial y}{\partial v} - \frac{\partial x}{\partial v} \frac{\partial y}{\partial u}$$

The Jacobian relates the area of any infinitesimal region inside D_0 with the corresponding region inside $D = G(D_0)$. In fact, if G is a linear map, then $\text{Jac}(G)$ is constant and is equal in magnitude to the ratio of the areas of D to that of D_0 :

Jacobian of a Linear Map: If $G(u, v) = (A u + C v, B u + D v)$ is a linear mapping from D_0 to D , then $\text{Jac}(G)$ is constant with value

$$\text{Jac}(G) = \begin{vmatrix} A & C \\ B & D \end{vmatrix} = A D - B C$$

Moreover,


$$\text{Area}(D) = |\text{Jac}(G)| \text{Area}(D_0)$$

Refer to your textbook for a detailed discussion of transformations of plane regions.

Example 15.12. Make an appropriate changes of variables to calculate the double integral $\iint_D x y \, dA$, where D is the region bounded by the curves $x y = 1$, $x y = 2$, $x^2 y^2 = 1$, and $x^2 y^2 = 2$.

Solution: Here is a plot of the shaded region D bounded by the four given curves:


```
In[495]:= plot1 = ContourPlot[{x * y == 1, x * y == 2, x * y^2 == 1, x * y^2 == 2},
  {x, 0, 5}, {y, 0, 5}, AspectRatio -> Automatic, ImageSize -> {250}];
plot2 = ContourPlot[1, {x, 0, 5}, {y, 0, 5}, AspectRatio -> Automatic,
  RegionFunction -> Function[{x, y}, 1 < x y < 2 && 1 < x y^2 < 2],
  ImageSize -> {250}, PlotPoints -> 100];
Show[
  plot1,
  plot2]
```



Observe that D is rather complicated. Since D can be described by the inequalities $1 < x y < 2$ and $1 < x^2 y^2 < 2$, we make the natural change of variables $u = x y$ and $v = x^2 y^2$, which transforms D to a simple square region D_0 in the uv -plane bounded by

$u = 1, u = 2, v = 1,$ and $v = 2$:

```
In[498]:= ContourPlot[1, {u, 0, 3}, {v, 0, 3}, ImageSize -> {250},
  RegionFunction -> Function[{u, v}, 1 < u < 2 && 1 < v < 2]]
```



To find the formula for our transformation $G(u, v) = (x(u, v), y(u, v))$ that maps D_0 to D , we solve for x and y in terms of u and v :

```
In[499]:= Clear[sol, x, y, u, v]
  sol = Solve[{u == x * y, v == x * y^2}, {x, y}]
```

Out[500]= $\left\{ \left\{ x \rightarrow \frac{u^2}{v}, y \rightarrow \frac{v}{u} \right\} \right\}$

It follows that $G(u, v) = (u^2/v, v/u)$ and the corresponding Jacobian is

```
In[501]:= x = sol[[1, 1, 2]]
  y = sol[[1, 2, 2]]
  Jac = D[x, u] * D[y, v] - D[x, v] * D[y, u]
```

Out[501]= $\frac{u^2}{v}$

Out[502]= $\frac{v}{u}$

Out[503]= $\frac{1}{v}$

Thus, the given integral transforms to $\iint_D x y \, dA = \iint_{D_0} \frac{u}{v} \, dA = \int_1^2 \int_1^2 \frac{u}{v} \, dv \, du$ with value

```
In[504]:= Integrate[u / v, {u, 1, 2}, {v, 1, 2}]
```

Out[504]= $\frac{3 \operatorname{Log}[2]}{2}$

■ Exercises

1. Consider the transformation $G(u, v) = (2u + v, u - 3v)$.

a. Set $D = G(D_0)$ where $D_0 = \{0 \leq u \leq 1, 0 \leq v \leq 2\}$. Make a plot of D and describe its shape.

b. Compute $\text{Jac}(G)$.

c. Compare the area of D with that of D_0 . How does this relate to $\text{Jac}(G)$?

2. Compute the area of the ellipse $\frac{x^2}{4} + \frac{y^2}{9} = 1$ by viewing it as a transformation of the unit circle $u^2 + v^2 = 1$ under a linear map $G(u, v) = (x(u, v), y(u, v))$ and using the area relationship described by $\text{Jac}(G)$.

3. Evaluate the integral $\iint_D x y \, dA$, where D is the region in the first quadrant bounded by the equations $y = x$, $y = 4x$, $xy = 1$, and $xy = 4$. HINT: Consider the change of variables $u = xy$ and $v = y$.

4. Evaluate the integral $\iint_D (x + y)/(x - y) \, dA$, where D is the parallelogram bounded by the lines $x - y = 1$, $x - y = 3$, $2x + y = 0$, and $2x + y = 2$. HINT: Consider the change of variables $u = x - y$ and $v = 2x + y$.

5. Evaluate the integral $\iint_D \frac{y}{x} \, dA$, where D is the region bounded by the circles $x^2 + y^2 = 1$, $x^2 + y^2 = 4$ and lines $y = x$, $y = 3x$. HINT: Consider the change of variables $u = x^2 + y^2$ and $v = y/x$.